Name : Abd-Alrhman Ali Alzboon

ID : 114147

```
*-----------------------------------------------------------
* Title     :       KNAPSACK PROBLEM
* Written by :     ABDALRHMAN ALZBOON
* Date      :       17/12/2017
* Description:     ...
*-----------------------------------------------------------

*CALCULATE THE MAXIMUM VALUE THAT CAN BE OBTAINED..
        ORG    $1000
START:
        MOVEQ       #0,D0   ;COUNTER MASK (SET OF OBJECT TO TAKE)
        MOVEQ       #0,D1   ;MAX SUM OF VALUES
        MOVEQ       #0,D2   ;SOLUTION MASK (OPTIMAL SET OF OBJECTS)
        MOVEQ       #0,D3   ;WEIGHT OF SOLUTION
OLOP    MOVEQ       #0,D4   ;SUM OF VALUES OF MASK(D0)
        MOVEQ       #0,D5   ;SUM OF WEIGHTS OF MASK(D0)
        MOVE.L      D0,D6
        LEA         V,A5
        LEA         W,A6
ILOP    BTST        #0,D6
        BEQ         NSET
        ADD.W       (A5),D4 ;ADD VALUE
        ADD.W       (A6),D5 ;ADD WEIGHT
        CMP.W       (K),D5
        BGT         OUT
NSET    ADDA        #2,A5
        ADDA        #2,A6
        ASR.L       #1,D6
        CMPI.W      #0,D6
        BNE         ILOP
        CMP.W       (K),D5
        BGT         OUT
        CMP.W       D1,D4
        BLE         OUT
        MOVE.L      D4,D1   ;UPD MAX VAL
        MOVE.L      D0,D2   ;UPD SET OF OBJ
        MOVE.L      D5,D3   ;UPD WEIGHT
OUT     ADDQ.L      #1,D0
        CMPI.L      #$10000,D0  ;COMPARE COUNTER WITH 2^16
        BLT         OLOP

        SIMHALT
*VARIABLES FOR WEIGHTS W,VALUES V ,WEIGHT CAPACITY K...
V       DC.W    74,51,74,53,74,70,38,47,42,64,30,47,64,42,52,70
W       DC.W    66,31,71,50,85,33,86,64,49,78,69,61,47,38,50,88
*V      DC.W    11,11,11,11,11,11,11,11,11,11,11,11,11,11,11,11
*W      DC.W    4,4,4,4,4,4,4,4,4,4,4,4,4,4,4,4
*V      DC.W    12,38,25,31,20,15,49,49,13,35,36,12,38,22,46,33
*W      DC.W    38,47,42,50,33,44,23,33,32,47,49,26,50,50,22,29
K       DC.W    400
        END     START
```

Approach

Solving Knapsack problem using Brute force search requires that we should try all combinations of choosing items and then maximizing the value of the items taken for each combination.

In order to do this,first we should observe that a combination can be presented as a binary string, where 0 means that the item at index (i) will not be taken, and 1 for items that will be taken, e.g “010110” the items at index {2,3,5} from right will be taken, the number of combinations is 2^{16} since we have 16 items, then for each combination we calculate the sum of values and the sum of weights then we maximize the final result..

The time complexity for this solution is O(2^N) where N is the number of items
The space complexity is O(N)..

Test case

Inputs :

Values = {74,51,74,53,74,70,38,47,42,64,30,47,64,42,52,70}
Weights = {66,31,71,50,85,33,86,64,49,78,69,61,47,38,50,88}
Capacity k = 400

Outputs :

Maximum value = 480
Set of items = {1,2,3,4,6,9,13,15}
Weight = 397